Matrix Convex Sets over the Euclidean Ball and polar duals of real free spectrahedra Theorem 2.9 verification Evert Passer.

Updated 29 July 2025.

Note: Some displays are cutoff in the PDF version due to large outputs. Please see the notebook version for full displays.

This notebook requires NCSpectrahedronExtreme to run.

Note: NCSpectrahedronExtreme can be installed by executing the Mathematica command

PacletInstall["https://www.wolframcloud.com/obj/ericmevert/NCSpectrahedronExtreme-3.1.2.paclet"]

This line is intentionally left as text so NCSpectrahedronExtreme will not install unless the user chooses to do so.

Initialization. This cell must be evaluated.

This section calls NCSpectrahedronExtreme and inputs short forms of the IdentityMatrix, Conjugate-Transpose, and KroneckerProduct functions.

```
In[1]:= << NCSpectrahedronExtreme`
id = IdentityMatrix;
ct = ConjugateTranspose;
kp = KroneckerProduct;
```

Defining Pencil for F[3]. This cell must be evaluated.

In[5]:=
```
P1 = {{1, 0}, {0, -1}};
P2 = {{0, 1}, {1, 0}};
F1 = kp[P1, P1];
F2 = kp[P2, P1];
F3 = kp[IdentityMatrix[2], P2];
F = {F1, F2, F3};
ViewTuple[F]
```

Out[11]=
$$\left\{\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \\ 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{pmatrix}\right\}$$

Extreme point at level 4.

Defining the tuple X and verifying it is self-adjoint. We will work with the real valued extreme point described in Remark 2.10.

In[12]:=
```
C1 = 1 / 2 * {{1 + 1 / Sqrt[3], 0}, {0, -1 + Sqrt[3]}};
Xc1 = ArrayFlatten[{{0, C1}, {Conjugate[C1], 0}}];
C2 = 1 / 2 * {{0, 1}, {1, 0}};
Xc2 = ArrayFlatten[{{0, C2}, {Conjugate[C2], 0}}];
C3 = 1 / 2 * {{-I + 2 I / Sqrt[3], 0}, {0, -I}};
Xc3 = ArrayFlatten[{{0, C3}, {Conjugate[C3], 0}}];
Xc = {Xc1, Xc2, Xc3} // FullSimplify;
ViewTuple[Xc]
```

Out[19]=

$$\left\{\begin{pmatrix} 0 & 0 & \frac{1}{6}\left(3+\sqrt{3}\right) & 0 \\ 0 & 0 & 0 & \frac{1}{2}\left(-1+\sqrt{3}\right) \\ \frac{1}{6}\left(3+\sqrt{3}\right) & 0 & 0 & 0 \\ 0 & \frac{1}{2}\left(-1+\sqrt{3}\right) & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & \frac{1}{2} \\ 0 & 0 & \frac{1}{2} & 0 \\ 0 & \frac{1}{2} & 0 & 0 \\ \frac{1}{2} & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & \frac{1}{6} i \left(-3+2\sqrt{3}\right) & 0 \\ 0 & 0 & 0 & -\frac{i}{2} \\ \frac{1}{6} i \left(3-2\sqrt{3}\right) & 0 & 0 & 0 \\ 0 & \frac{i}{2} & 0 & 0 \end{pmatrix}\right\}$$

The following line computes the norms of Xi-ConjugateTranspose[Xi].

In[20]:=
```
Map[Norm, Xc - Map[ct, Xc] // Simplify]
```

Out[20]= {0, 0, 0}

To obtain the real tuple from Remark 2.10, we conjugate Xc by the unitary U. We first verify U is unitary.

In[21]:=
```
U = Sqrt[2] / 2 * ArrayFlatten[{{id[2], -I * id[2]}, {id[2], I * id[2]}}];
Norm[IdentityMatrix[4] - ConjugateTranspose[U].U]
```

Out[22]= 0

We now conjugate X by U.

In[23]:=
```
X = {X1, X2, X3} = {ct[U].Xc1.U, ct[U].Xc2.U, ct[U].Xc3.U} // Simplify;
X // ViewTuple
```

Out[24]=
$$\left\{\begin{pmatrix} \frac{1}{6}\left(3+\sqrt{3}\right) & 0 & 0 & 0 \\ 0 & \frac{1}{2}\left(-1+\sqrt{3}\right) & 0 & 0 \\ 0 & 0 & \frac{1}{6}\left(-3-\sqrt{3}\right) & 0 \\ 0 & 0 & 0 & \frac{1}{2}\left(1-\sqrt{3}\right) \end{pmatrix},\right.$$
$$\left.\begin{pmatrix} 0 & \frac{1}{2} & 0 & 0 \\ \frac{1}{2} & 0 & 0 & 0 \\ 0 & 0 & 0 & -\frac{1}{2} \\ 0 & 0 & -\frac{1}{2} & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & \frac{1}{2}-\frac{1}{\sqrt{3}} & 0 \\ 0 & 0 & 0 & \frac{1}{2} \\ \frac{1}{2}-\frac{1}{\sqrt{3}} & 0 & 0 & 0 \\ 0 & \frac{1}{2} & 0 & 0 \end{pmatrix}\right\}$$

We verify that the form presented in the article is the same as the form here by comparing the key entries.

In[25]:=
```
alpha = 1 + 1 / Sqrt[3];
```
$$\frac{1}{6}\left(3+\sqrt{3}\right) == \frac{1}{2}\,\text{alpha}$$
$$\frac{1}{2}\left(-1+\sqrt{3}\right) == \frac{1}{2}\,(3\,\text{alpha}-4)$$
$$\frac{1}{2}-\frac{1}{\sqrt{3}} == \frac{1}{2}\,(3-2\,\text{alpha})$$

Out[26]= True

Out[27]= True

Out[28]= True

Verification that X is an element of the free spectrahedron D_F.

The command LMI[F,X] computes L_F(X) as defined in the article. We need only check that this is positive semidefinite. We can also use the mathematica command PositiveSemidefiniteMatrixQ

In[29]:=
```
Eigenvalues[LMI[F, X]] // Simplify
PositiveSemidefiniteMatrixQ[LMI[F, X]]
```

Out[29]=
$$\left\{2, 2, 2, 2, 2, 2, \frac{2}{\sqrt{3}}, \frac{2}{\sqrt{3}}, 2-\frac{2}{\sqrt{3}}, 2-\frac{2}{\sqrt{3}}, 0, 0, 0, 0, 0, 0\right\}$$

Out[30]= True

Verification that X is irreducible.

It is sufficient to show that the only matrices that commute with each entry of X are constant multiples of the idenity. We first do this by solving all equations at once.

In[31]:=
```
Wmat = Table[W[i, j], {i, 4}, {j, 4}];
WmatrusAll = Solve[
   Wmat.X1 - X1.Wmat == 0 && Wmat.X2 - X2.Wmat == 0 && Wmat.X3 - X3.Wmat == 0, Variables[Wmat]];
(* The length of WmatrusAll is computed to verify that there
 is no alternative solution. *)
Length[WmatrusAll]
Wmat //. WmatrusAll[[1]] // MatrixForm
```

Solve: Equations may not give solutions for all "solve" variables.

Out[33]= 1

Out[34]//MatrixForm=

$$\begin{pmatrix} W[1,1] & 0 & 0 & 0 \\ 0 & W[1,1] & 0 & 0 \\ 0 & 0 & W[1,1] & 0 \\ 0 & 0 & 0 & W[1,1] \end{pmatrix}$$

We next sequentially find matrices that commute with each individual Xi. We start with X1.

In[35]:=
```
Wmatrus = Solve[Wmat.X1 - X1.Wmat == 0];
Length[Wmatrus]
Wmat //. Wmatrus[[1]] // MatrixForm
Wmat1 = Wmat //. Wmatrus[[1]];
```

Out[36]= 1

Out[37]//MatrixForm=

$$\begin{pmatrix} W[1,1] & 0 & 0 & 0 \\ 0 & W[2,2] & 0 & 0 \\ 0 & 0 & W[3,3] & 0 \\ 0 & 0 & 0 & W[4,4] \end{pmatrix}$$

We now find the W with the above form that commute with X2

In[39]:=
```
Wmat1rus = Solve[Wmat1.X2 - X2.Wmat1 == 0];
Length[Wmat1rus]
Wmat2 = Wmat1 //. Wmat1rus[[1]];
Wmat2 // MatrixForm
```

Out[40]= 1

Out[42]//MatrixForm=

$$\begin{pmatrix} W[1,1] & 0 & 0 & 0 \\ 0 & W[1,1] & 0 & 0 \\ 0 & 0 & W[3,3] & 0 \\ 0 & 0 & 0 & W[3,3] \end{pmatrix}$$

In[43]:=
```
(* Finally, if W is such a matrix that commutes with X3,
then W must be a constant multiple of the identity *)
```

In[44]:=
```
Wmat2rus = Solve[Wmat2.X3 - X3.Wmat2 == 0];
Length[Wmat2rus]
Wmat3 = Wmat2 //. Wmat2rus[[1]];
Wmat3 // MatrixForm
```

Out[45]= 1

Out[47]//MatrixForm=

$$\begin{pmatrix} W[1,1] & 0 & 0 & 0 \\ 0 & W[1,1] & 0 & 0 \\ 0 & 0 & W[1,1] & 0 \\ 0 & 0 & 0 & W[1,1] \end{pmatrix}$$

Proof that X is Arveson Extreme Point. Done by computing a matrix representation of the map Lambda[F,\cdot].NullBasisMat and computing showing the null space of the map is trivial. This is equivalent to the statement that there is no beta that nontrivially dilates X, i.e., that X is Arveson extreme.

We first use the NCSpectrahedronExtremeCommand ArvesonTest to check that X is ArvesonExtreme. If the Option NumericalQ is set to False this command will use algebraic arithmetic to verify the point is extreme. Below “True” states that the point is ArvesonExtreme. The remaining values describe how robustly the appropriate null spaces were found.

In[48]:=
```
ArvesonTest[F, X, NumericalQ → False]
```

Out[48]=

$$\left\{\text{True}, \left\{\left\{0, \sqrt{\frac{1.04\ldots}{3.77\ldots}}\right\}, \left\{\sqrt{31.3\ldots}, \sqrt{31.3\ldots}, \sqrt{19.5\ldots}, \sqrt{19.5\ldots}, \sqrt{13.9\ldots}, \sqrt{13.9\ldots}, \sqrt{10.4\ldots}, \sqrt{10.4\ldots}, \sqrt{3.77\ldots}, \sqrt{3.77\ldots}, \sqrt{1.04\ldots}, \sqrt{1.04\ldots}\right\}\right\}\right\}$$

Mathematica has algebraically represented these numbers are roots of polynomials. An example of this is below

In[ ]:=
```
Sqrt[Root[{-3 + #1^2 &, -1024 + 1056 #2 + 32 #1 #2 - 118 #2^2 - 6 #1 #2^2 + 3 #2^3 &}, {2, 3}]] ==
  Sqrt[31.3...]
```

Out[ ]= True

We additionally go through the computation in a step by step manor so the user can see the steps more carefully.

ArvesonTestSystemExact below computes a basis for the nullspace of LMI[F,X] and moreover computes a matrix representation of the map Lambda[F,\cdot].NullBasisMat

In[49]:=
```
{ArvMat, NullBasisMat} = ArvesonTestSystemExact[F, X] // FullSimplify;
```

We verify that the nullspace basis was computed correctly by
1. Multiplying LMI[F,X].NullBasisMat.
2. Finding the dimension of the null space of LMI[F,X] by counting the number of eigenvalues of LMI[F,X] equal to 0 (recall LMI[F,X] is self-adjiont).
3. Showing NullBasisMat has full rank.

1. Multiplying LMI[F,X].NullBasisMat.

In[50]:=
```
Norm[LMI[F, X].NullBasisMat // FullSimplify]
```

Out[50]= 0

2. Finding the dimension of the null space of LMI[F,X] by counting the number of eigenvalues of LMI[F,X] equal to 0 (recall LMI[F,X] is self-adjiont).

In[51]:=
```
Dimensions[NullBasisMat]
Count[Eigenvalues[LMI[F, X]], 0]
```

Out[51]= {16, 6}

Out[52]= 6

3. Showing NullBasisMat has full rank. Mathematica can compute the singular values.

In[53]:=
```
SingularValueList[NullBasisMat] // FullSimplify
Length[SingularValueList[NullBasisMat] // FullSimplify]
```

Out[53]= $$\left\{2\left(1+\sqrt{3}\right),\ 2\sqrt{2+\sqrt{3}},\ 2,\ 2\left(-1+\sqrt{3}\right),\ \frac{2}{\sqrt{3}},\ \sqrt{2}\left(-1+\sqrt{3}\right)\right\}$$

Out[54]= 6

It is also easy to see this matrix has full rank by looking at the submatrix formed from the rows 10, 11, 13, 14, 15, 16.

In[55]:=
```
NullBasisMat[[{10, 11, 13, 14, 15, 16}]] // MatrixForm
```

Out[55]//MatrixForm=
$$\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

ArvMat is the matrix representation of the map Lambda[F,\cdot].NullBasisMat. It remains to show the matrix has full rank. We can do this again by computing singular values.

```
In[56]:= SingularValueList[ArvMat] // FullSimplify
Length[SingularValueList[ArvMat] // FullSimplify]
Dimensions[ArvMat]
```

Out[56]= $\{\sqrt{31.3\ldots}, \sqrt{31.3\ldots}, \sqrt{19.5\ldots}, \sqrt{19.5\ldots}, \sqrt{13.9\ldots}, \sqrt{13.9\ldots}, \sqrt{10.4\ldots}, \sqrt{10.4\ldots}, \sqrt{3.77\ldots}, \sqrt{3.77\ldots}, \sqrt{1.04\ldots}, \sqrt{1.04\ldots}\}$

Out[57]= 12

Out[58]= {24, 12}

Numerical approximations of the singular values can be computed as follows.

```
In[60]:= SingularValueList[ArvMat] // N
```

Out[60]= {5.59597, 5.59597, 4.42058, 4.42058, 3.72664, 3.72664, 3.23084, 3.23084, 1.94246, 1.94246, 1.02188, 1.02188}

The above verifies that ArvMat has full rank so long as one trusts the algebraic singular value computation of Mathematica. An alternative approach follows.

We observe that up to permutations, ArvMat can be written as a direct sum of four matrices. We can then show that each submatrix has full rank by computing the determinant of an appropriate submatrix.

The permutations we need are permL and permR below.

```
In[61]:= RowPermIndicies =
  {1, 6, 9, 14, 22, 23, 2, 10, 11, 13, 18, 21, 3, 7, 12, 16, 17, 20, 4, 5, 8, 15, 19, 24};
ColPermIndicies = {1, 6, 11, 2, 5, 12, 3, 8, 9, 4, 7, 10};
permL = Transpose[Permute[IdentityMatrix[24], RowPermIndicies]];
permR = Transpose[Permute[IdentityMatrix[12], ColPermIndicies]];
```

After left and right multiplying by the permutation matrices, ArvMat indeed has a block structure.

In[65]:=
```
ArvMatPerm = permL.((ArvMat).Transpose[permR]);
Dimensions[ArvMatPerm]
ArvMatPerm // MatrixForm
```

Out[66]= {24, 12}

Out[67]//MatrixForm=

$$
\begin{pmatrix}
-1 & 0 & 1+\sqrt{3} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
1+\sqrt{3} & 1 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
2-\sqrt{3} & -1 & -2+\sqrt{3} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
-1 & 2+\sqrt{3} & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
1 & 1-\frac{1}{\sqrt{3}} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -1+\frac{2}{\sqrt{3}} & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 2+\sqrt{3} & 1 & 2+\sqrt{3} & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 1-\frac{1}{\sqrt{3}} & -1 & 1-\frac{2}{\sqrt{3}} & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & -1+\frac{2}{\sqrt{3}} & 0 & -1+\frac{1}{\sqrt{3}} & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & -1 & 1 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & -1 & 1+\sqrt{3} & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 1 & 2-\sqrt{3} & 1 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & -2+\sqrt{3} & 1 & -2+\sqrt{3} & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & -1-\sqrt{3} & -1 & -1 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 1+\sqrt{3} & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 1-\frac{2}{\sqrt{3}} & 1 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & -1 & -1+\frac{1}{\sqrt{3}} & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 1 & -2-\sqrt{3} & 1 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1-\frac{2}{\sqrt{3}} & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1+\frac{1}{\sqrt{3}} & 1 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -2-\sqrt{3} & -1 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & -2+\sqrt{\ } \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & -1-\sqrt{\ } \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1
\end{pmatrix}
$$

In[68]:=
```
ArvMatPermB1 = ArvMatPerm[[1 ;; 6, 1 ;; 3]];
ArvMatPermB2 = ArvMatPerm[[7 ;; 12, 4 ;; 6]];
ArvMatPermB3 = ArvMatPerm[[13 ;; 18, 7 ;; 9]];
ArvMatPermB4 = ArvMatPerm[[19 ;; 24, 10 ;; 12]];
```

In[72]:=
```
ArvMatPermB1 // MatrixForm
ArvMatPermB2 // MatrixForm
ArvMatPermB3 // MatrixForm
ArvMatPermB4 // MatrixForm
```

Out[72]//MatrixForm=

$$\begin{pmatrix} -1 & 0 & 1+\sqrt{3} \\ 1+\sqrt{3} & 1 & -1 \\ 2-\sqrt{3} & -1 & -2+\sqrt{3} \\ -1 & 2+\sqrt{3} & 1 \\ 1 & 1-\frac{1}{\sqrt{3}} & 0 \\ 0 & -1+\frac{2}{\sqrt{3}} & 1 \end{pmatrix}$$

Out[73]//MatrixForm=

$$\begin{pmatrix} 2+\sqrt{3} & 1 & 2+\sqrt{3} \\ 1-\frac{1}{\sqrt{3}} & -1 & 1-\frac{2}{\sqrt{3}} \\ -1+\frac{2}{\sqrt{3}} & 0 & -1+\frac{1}{\sqrt{3}} \\ 0 & -1 & 1 \\ -1 & 1+\sqrt{3} & 0 \\ 1 & 2-\sqrt{3} & 1 \end{pmatrix}$$

Out[74]//MatrixForm=

$$\begin{pmatrix} -2+\sqrt{3} & 1 & -2+\sqrt{3} \\ -1-\sqrt{3} & -1 & -1 \\ 1 & 0 & 1+\sqrt{3} \\ 0 & 1-\frac{2}{\sqrt{3}} & 1 \\ -1 & -1+\frac{1}{\sqrt{3}} & 0 \\ 1 & -2-\sqrt{3} & 1 \end{pmatrix}$$

Out[75]//MatrixForm=

$$\begin{pmatrix} 1-\frac{2}{\sqrt{3}} & 0 & -1+\frac{1}{\sqrt{3}} \\ -1+\frac{1}{\sqrt{3}} & 1 & 1-\frac{2}{\sqrt{3}} \\ -2-\sqrt{3} & -1 & 2+\sqrt{3} \\ -1 & -2+\sqrt{3} & 1 \\ 1 & -1-\sqrt{3} & 0 \\ 0 & 1 & 1 \end{pmatrix}$$

We verify that we have obtained the correct blocks by checking that ArvMatPerm is equal to the direct sum of the blocks.

In[76]:=
```
ArvMatPerm == DirectSum[{ArvMatPermB1, ArvMatPermB2, ArvMatPermB3, ArvMatPermB4}]
```

Out[76]= True

Now for each block we compute the determinant of the submatrix formed from the first three rows. In all cases, the determinant is nonzero, hence each block has full rank, so ArvMat indeed has full rank, which completes the proof.

In[77]:=
```
Det[ArvMatPermB1[[1 ;; 3]]] // FullSimplify
Det[ArvMatPermB2[[1 ;; 3]]] // FullSimplify
Det[ArvMatPermB3[[1 ;; 3]]] // FullSimplify
Det[ArvMatPermB4[[1 ;; 3]]] // FullSimplify
```

Out[77]= $-4\sqrt{3}$

Out[78]= $\frac{4}{\sqrt{3}}$

Out[79]= $4\sqrt{3}$

Out[80]= $-\frac{4}{\sqrt{3}}$

Extreme point at level 6.

Defining the tuple X and verifying it is self-adjoint. We again work with a real tuple that is unitarily equivalent to the tuple described in Theorem 2.9.

In[81]:=
```
al = Sqrt[2] - 1;
C1 = 1 / 4 * {{al + 1, 0, -al}, {0, al + 1, -al}, {-al, -al, al + 1}};
Xc1 = ArrayFlatten[{{0, C1}, {Conjugate[C1], 0}}];
C2 = 1 / 4 * {{-4 * Sqrt[2 al], 0, 0}, {0, 4 * Sqrt[2 al], 0}, {0, 0, 0}};
Xc2 = ArrayFlatten[{{0, C2}, {Conjugate[C2], 0}}];
C3 = I / 4 * {{0, 3 al - 1, al}, {3 al - 1, 0, al}, {al, al, 3 - al}};
Xc3 = ArrayFlatten[{{0, C3}, {Conjugate[C3], 0}}];
Xc = {Xc1, Xc2, Xc3} // Simplify;
ViewTuple[Xc]
```

Out[89]=

$$\left\{\begin{pmatrix} 0 & 0 & 0 & \frac{1}{2\sqrt{2}} & 0 & \frac{1}{4}\left(1-\sqrt{2}\right) \\ 0 & 0 & 0 & 0 & \frac{1}{2\sqrt{2}} & \frac{1}{4}\left(1-\sqrt{2}\right) \\ 0 & 0 & 0 & \frac{1}{4}\left(1-\sqrt{2}\right) & \frac{1}{4}\left(1-\sqrt{2}\right) & \frac{1}{2\sqrt{2}} \\ \frac{1}{2\sqrt{2}} & 0 & \frac{1}{4}\left(1-\sqrt{2}\right) & 0 & 0 & 0 \\ 0 & \frac{1}{2\sqrt{2}} & \frac{1}{4}\left(1-\sqrt{2}\right) & 0 & 0 & 0 \\ \frac{1}{4}\left(1-\sqrt{2}\right) & \frac{1}{4}\left(1-\sqrt{2}\right) & \frac{1}{2\sqrt{2}} & 0 & 0 & 0 \end{pmatrix},\right.$$

$$\begin{pmatrix} 0 & 0 & 0 & -\sqrt{2\left(-1+\sqrt{2}\right)} & 0 & 0 \\ 0 & 0 & 0 & 0 & \sqrt{2\left(-1+\sqrt{2}\right)} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ -\sqrt{2\left(-1+\sqrt{2}\right)} & 0 & 0 & 0 & 0 & 0 \\ 0 & \sqrt{2\left(-1+\sqrt{2}\right)} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & \\ 0 & \\ 0 & \\ 0 & -\frac{1}{4} \\ -\frac{1}{4}\,i\left(-4+3\sqrt{2}\right) & \\ -\frac{1}{4}\,i\left(-1+\sqrt{2}\right) & -\frac{1}{4} \end{pmatrix}$$

The following line computes the norms of Xi-ConjugateTranspose[Xi].

In[90]:=
```
Map[Norm, Xc - Map[ct, Xc] // Simplify]
```

Out[90]= {0, 0, 0}

We obtain a real tuple by first conjugating by a generalization of the unitary found in Remark 2.10. We then conjugate by a second unitary so that X1 is diagonal in the resulting tuple. This ends up being easier to do computations with later.

In[91]:=
```
U = Sqrt[2] / 2 * ArrayFlatten[{{id[3], -I * id[3]}, {id[3], I * id[3]}}];
Norm[IdentityMatrix[6] - ConjugateTranspose[U].U]
Xb = {Xb1, Xb2, Xb3} = {ct[U].Xc1.U, ct[U].Xc2.U, ct[U].Xc3.U} // Simplify;
```

Out[92]= 0

In[94]:=
```
Xb // ViewTuple
V = Transpose[Eigenvectors[Xb[[1, 1 ;; 3, 1 ;; 3]]] // Orthogonalize];
VV = DirectSum[{V, V}];
Norm[IdentityMatrix[6] - ConjugateTranspose[VV].VV]
```

Out[94]=

$$\left\{\begin{pmatrix} \frac{1}{2\sqrt{2}} & 0 & \frac{1}{4}\left(1-\sqrt{2}\right) & 0 & 0 & 0 \\ 0 & \frac{1}{2\sqrt{2}} & \frac{1}{4}\left(1-\sqrt{2}\right) & 0 & 0 & 0 \\ \frac{1}{4}\left(1-\sqrt{2}\right) & \frac{1}{4}\left(1-\sqrt{2}\right) & \frac{1}{2\sqrt{2}} & 0 & 0 & 0 \\ 0 & 0 & 0 & -\frac{1}{2\sqrt{2}} & 0 & \frac{1}{4}\left(-1+\sqrt{2}\right) \\ 0 & 0 & 0 & 0 & -\frac{1}{2\sqrt{2}} & \frac{1}{4}\left(-1+\sqrt{2}\right) \\ 0 & 0 & 0 & \frac{1}{4}\left(-1+\sqrt{2}\right) & \frac{1}{4}\left(-1+\sqrt{2}\right) & -\frac{1}{2\sqrt{2}} \end{pmatrix},\right.$$

$$\begin{pmatrix} -\sqrt{2\left(-1+\sqrt{2}\right)} & 0 & 0 & 0 & 0 & 0 \\ 0 & \sqrt{2\left(-1+\sqrt{2}\right)} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & \sqrt{2\left(-1+\sqrt{2}\right)} & 0 & 0 \\ 0 & 0 & 0 & 0 & -\sqrt{2\left(-1+\sqrt{2}\right)} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix},$$

$$\left.\begin{pmatrix} 0 & 0 & 0 & 0 & 1-\frac{3}{2\sqrt{2}} & \frac{1}{4}\left(1-\sqrt{2}\right) \\ 0 & 0 & 0 & 1-\frac{3}{2\sqrt{2}} & 0 & \frac{1}{4}\left(1-\sqrt{2}\right) \\ 0 & 0 & 0 & \frac{1}{4}\left(1-\sqrt{2}\right) & \frac{1}{4}\left(1-\sqrt{2}\right) & \frac{1}{4}\left(-4+\sqrt{2}\right) \\ 0 & 1-\frac{3}{2\sqrt{2}} & \frac{1}{4}\left(1-\sqrt{2}\right) & 0 & 0 & 0 \\ 1-\frac{3}{2\sqrt{2}} & 0 & \frac{1}{4}\left(1-\sqrt{2}\right) & 0 & 0 & 0 \\ \frac{1}{4}\left(1-\sqrt{2}\right) & \frac{1}{4}\left(1-\sqrt{2}\right) & \frac{1}{4}\left(-4+\sqrt{2}\right) & 0 & 0 & 0 \end{pmatrix}\right\}$$

Out[97]= 0

We now conjugate Xb to diagonalize the frist entry.

In[98]:=
```
X = {X1, X2, X3} = {ct[VV].Xb1.VV, ct[VV].Xb2.VV, ct[VV].Xb3.VV} // FullSimplify;
X // ViewTuple
```

Out[99]=
$$\left\{\begin{pmatrix} \frac{1}{2} & 0 & 0 & 0 & 0 & 0 \\ 0 & \frac{1}{2\sqrt{2}} & 0 & 0 & 0 & 0 \\ 0 & 0 & -\frac{1}{2}+\frac{1}{\sqrt{2}} & 0 & 0 & 0 \\ 0 & 0 & 0 & -\frac{1}{2} & 0 & 0 \\ 0 & 0 & 0 & 0 & -\frac{1}{2\sqrt{2}} & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{1}{2}-\frac{1}{\sqrt{2}} \end{pmatrix},\right.$$

$$\begin{pmatrix} 0 & -\sqrt{-1+\sqrt{2}} & 0 & 0 & 0 & 0 \\ -\sqrt{-1+\sqrt{2}} & 0 & \sqrt{-1+\sqrt{2}} & 0 & 0 & 0 \\ 0 & \sqrt{-1+\sqrt{2}} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & \sqrt{-1+\sqrt{2}} & 0 \\ 0 & 0 & 0 & \sqrt{-1+\sqrt{2}} & 0 & -\sqrt{-1+\sqrt{2}} \\ 0 & 0 & 0 & 0 & -\sqrt{-1+\sqrt{2}} & 0 \end{pmatrix},$$

$$\left.\begin{pmatrix} 0 & 0 & 0 & \frac{1}{2}-\frac{1}{\sqrt{2}} & 0 & -1+\frac{1}{\sqrt{2}} \\ 0 & 0 & 0 & 0 & -1+\frac{3}{2\sqrt{2}} & 0 \\ 0 & 0 & 0 & -1+\frac{1}{\sqrt{2}} & 0 & -\frac{1}{2} \\ \frac{1}{2}-\frac{1}{\sqrt{2}} & 0 & -1+\frac{1}{\sqrt{2}} & 0 & 0 & 0 \\ 0 & -1+\frac{3}{2\sqrt{2}} & 0 & 0 & 0 & 0 \\ -1+\frac{1}{\sqrt{2}} & 0 & -\frac{1}{2} & 0 & 0 & 0 \end{pmatrix}\right\}$$

Verification that X is an element of the free spectrahedron D_F.

The command LMI[F,X] computes L_F(X) as defined in the article. We need only check that this is positive semidefinite. We can also use the mathematica command PositiveSemidefiniteMatrixQ

In[100]:=
```
Eigenvalues[LMI[F, X]] // Simplify
PositiveSemidefiniteMatrixQ[LMI[F, X]]
```

Out[100]=
$$\left\{2, 2, 2, 2, 2, 2, 2, 2, 2, 2, 2-\frac{1}{\sqrt{2}}, 2-\frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, \frac{1}{\sqrt{2}}, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0\right\}$$

Out[101]= True

Verification that X is irreducible.

It is sufficient to show that the only matrices that commute with each entry of X are constant multiples of the idenity. We first do this by solving all equations at once.

In[102]:=
```
Wmat = Table[W[i, j], {i, 6}, {j, 6}];
WmatrusAll = Solve[
   Wmat.X1 - X1.Wmat == 0 && Wmat.X2 - X2.Wmat == 0 && Wmat.X3 - X3.Wmat == 0, Variables[Wmat]];
(* The length of WmatrusAll is computed to verify that there
 is no alternative solution. *)
Length[WmatrusAll]
Wmat //. WmatrusAll[[1]] // MatrixForm
```

Solve: Equations may not give solutions for all "solve" variables.

Out[104]= 1

Out[105]//MatrixForm=

$$\begin{pmatrix} W[2,2] & 0 & 0 & 0 & 0 & 0 \\ 0 & W[2,2] & 0 & 0 & 0 & 0 \\ 0 & 0 & W[2,2] & 0 & 0 & 0 \\ 0 & 0 & 0 & W[2,2] & 0 & 0 \\ 0 & 0 & 0 & 0 & W[2,2] & 0 \\ 0 & 0 & 0 & 0 & 0 & W[2,2] \end{pmatrix}$$

We next sequentially find matrices that commute with each individual Xi. We start with X1.

In[106]:=
```
Wmatrus = Solve[Wmat.X1 - X1.Wmat == 0];
Length[Wmatrus]
Wmat //. Wmatrus[[1]] // MatrixForm
Wmat1 = Wmat //. Wmatrus[[1]];
```

Out[107]= 1

Out[108]//MatrixForm=

$$\begin{pmatrix} W[1,1] & 0 & 0 & 0 & 0 & 0 \\ 0 & W[2,2] & 0 & 0 & 0 & 0 \\ 0 & 0 & W[3,3] & 0 & 0 & 0 \\ 0 & 0 & 0 & W[4,4] & 0 & 0 \\ 0 & 0 & 0 & 0 & W[5,5] & 0 \\ 0 & 0 & 0 & 0 & 0 & W[6,6] \end{pmatrix}$$

We now find the W with the above form that commute with X2

In[110]:=
```
Wmat1rus = Solve[Wmat1.X2 - X2.Wmat1 == 0];
Length[Wmat1rus]
Wmat2 = Wmat1 //. Wmat1rus[[1]];
Wmat2 // MatrixForm
```

Out[111]= 1

Out[113]//MatrixForm=

$$\begin{pmatrix} W[2,2] & 0 & 0 & 0 & 0 & 0 \\ 0 & W[2,2] & 0 & 0 & 0 & 0 \\ 0 & 0 & W[2,2] & 0 & 0 & 0 \\ 0 & 0 & 0 & W[5,5] & 0 & 0 \\ 0 & 0 & 0 & 0 & W[5,5] & 0 \\ 0 & 0 & 0 & 0 & 0 & W[5,5] \end{pmatrix}$$

In[114]:=
```
(* Finally, if W is such a matrix that commutes with X3,
then W must be a constant multiple of the identity *)
```

In[115]:=
```
Wmat2rus = Solve[Wmat2.X3 - X3.Wmat2 == 0];
Length[Wmat2rus]
Wmat3 = Wmat2 //. Wmat2rus[[1]];
Wmat3 // MatrixForm
```

Out[116]= 1

Out[118]//MatrixForm=

$$\begin{pmatrix} W[2,2] & 0 & 0 & 0 & 0 & 0 \\ 0 & W[2,2] & 0 & 0 & 0 & 0 \\ 0 & 0 & W[2,2] & 0 & 0 & 0 \\ 0 & 0 & 0 & W[2,2] & 0 & 0 \\ 0 & 0 & 0 & 0 & W[2,2] & 0 \\ 0 & 0 & 0 & 0 & 0 & W[2,2] \end{pmatrix}$$

Proof that X is Arveson Extreme Point. Done by computing a matrix representation of the map Lambda[F,\cdot].NullBasisMat and computing showing the null space of the map is trivial. This is equivalent to the statement that there is no beta that nontrivially dilates X, i.e., that X is Arveson extreme.

The algebra in the computation in this case ends up being nicer if we use an alternative defining tuple A that is

unitarily equivalent to F. In particular, the basis for the null space ends up being nicer, which heavily impacts the remaining steps.

In[119]:=
```
A = {A1, A2, A3} = {{{-1, 0, 0, 0}, {0, 1, 0, 0}, {0, 0, -1, 0}, {0, 0, 0, 1}},
    {{0, 1, 0, 0}, {1, 0, 0, 0}, {0, 0, 0, -1}, {0, 0, -1, 0}},
    {{0, 0, 0, 1}, {0, 0, 1, 0}, {0, 1, 0, 0}, {1, 0, 0, 0}}};
A // ViewTuple
F // ViewTuple
```

Out[120]= $\left\{\begin{pmatrix} -1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & -1 \\ 0 & 0 & -1 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{pmatrix}\right\}$

Out[121]= $\left\{\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \\ 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \\ 1 & 0 & 0 & 0 \\ 0 & -1 & 0 & 0 \end{pmatrix}, \begin{pmatrix} 0 & 1 & 0 & 0 \\ 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 1 & 0 \end{pmatrix}\right\}$

In[122]:= $V = \left\{\left\{0, \frac{1}{\sqrt{2}}, 0, -\frac{1}{\sqrt{2}}\right\}, \left\{-\frac{1}{\sqrt{2}}, 0, \frac{1}{\sqrt{2}}, 0\right\}, \left\{\frac{1}{\sqrt{2}}, 0, \frac{1}{\sqrt{2}}, 0\right\}, \left\{0, \frac{1}{\sqrt{2}}, 0, \frac{1}{\sqrt{2}}\right\}\right\};$

```
Norm[ct[V].V - id[4]]
```

Out[123]= 0

In[124]:=
```
Map[Norm, A - {ct[V].F1.V, ct[V].F2.V, ct[V].F3.V}]
```

Out[124]= {0, 0, 0}

We first use the NCSpectrahedronExtremeCommand ArvesonTest to check that X is ArvesonExtreme. In this case, Mathematica struggles to do the exact arithmetic computation, so we perform the check numerically.

In[125]:=
```
ArvesonTest[A, X, NumericalQ → True]
```

Out[125]= {True, {{0, 0.302419}, {48.9822, 48.9822, 48.982, 48.982, 41.2254, 41.1963, 27.9221, 27.8202, 9.22715, 9.22376, 8.74655, 8.7438, 2.6443, 2.53671, 1.83759, 1.80274, 1.80079, 1.7818}}}

We now go through an exact computation similar to the one used in the n=4 case to prove that X is Arveson extreme.

ArvesonTestSystemExact below computes a basis for the nullspace of LMI[F,X] and moreover computes a matrix representation of the map Lambda[F,\cdot].NullBasisMat

In[126]:=
```
{ArvMat, NullBasisMat} = ArvesonTestSystemExact[A, X] // FullSimplify;
```

We verify that the nullspace basis was computed correctly by
1. Multiplying LMI[F,X].NullBasisMat.
2. Finding the dimension of the null space of LMI[F,X] by counting the number of eigenvalues of LMI[F,X]

equal to 0 (recall LMI[F,X] is self-adjiont).

3. Showing NullBasisMat has full rank.

1. Multiplying LMI[F,X].NullBasisMat.

```
In[127]:= Norm[LMI[A, X].NullBasisMat // FullSimplify]
Out[127]= 0
```

2. Finding the dimension of the null space of LMI[F,X] by counting the number of eigenvalues of LMI[F,X] equal to 0 (recall LMI[F,X] is self-adjiont).

```
In[128]:= Dimensions[NullBasisMat]
          Count[Eigenvalues[LMI[A, X]], 0]
Out[128]= {24, 10}
Out[129]= 10
```

3. Showing NullBasisMat has full rank. Mathematica now struggles to compute the singular values, so we consider a submatrix. The submatrix formed from the rows 24, 23, 22, 21, 20, 19, 18, 16, 15, 14 is equal to the 10 x 10 identity.

```
In[130]:= IdentityMatrix[10] == NullBasisMat[[{24, 23, 22, 21, 20, 19, 18, 16, 15, 14}]]
Out[130]= True
```

ArvMat is the matrix representation of the map Lambda[F,\cdot].NullBasisMat. In this case Mathematica struggles to compute the singular values algebraically so we perform the computation numerically. We also use the permutation matrix approach as in the n=4 case to prove the matrix has full rank.

```
In[131]:= SingularValueList[ArvMat // N] // FullSimplify
          Length[SingularValueList[ArvMat // N] // FullSimplify]
          Dimensions[ArvMat]
Out[131]= {48.9822, 48.9822, 48.982, 48.982, 41.2262, 41.1963, 27.9221, 27.8202, 9.23079,
           9.22738, 8.74655, 8.74383, 2.64505, 2.53667, 1.83759, 1.80266, 1.8008, 1.78181}
Out[132]= 18
Out[133]= {40, 18}
```

We observe that up to permutations, ArvMat can be written as a direct sum of four matrices. We can then show that each submatrix has full rank by computing the determinant of an appropriate submatrix.

The permutations we need are permL and permR below.

```
In[134]:= RowPermIndicies = {1, 3, 12, 17, 18, 25, 29, 34, 36, 40, 2, 7, 8, 11, 13, 24, 26, 30, 35,
             39, 4, 6, 10, 15, 19, 22, 27, 28, 31, 33, 5, 9, 14, 16, 20, 21, 23, 32, 37, 38};
          ColPermIndicies = {1, 3, 8, 16, 18, 2, 7, 9, 17, 4, 6, 11, 13, 15, 5, 10, 12, 14};
          permL = Transpose[Permute[IdentityMatrix[40], RowPermIndicies]];
          permR = Transpose[Permute[IdentityMatrix[18], ColPermIndicies]];
```

After left and right multiplying by the permutation matrices, ArvMat indeed has a block structure.

In[138]:=
```
ArvMatPerm = permL.((ArvMat).Transpose[permR]);
Dimensions[ArvMatPerm]
ArvMatPerm // MatrixForm
```

Out[139]= {40, 18}

Out[140]//MatrixForm=

$$
\left(\begin{array}{cccccc}
-\sqrt{2} & 1+\sqrt{2} & -\sqrt{7+5\sqrt{2}} & 0 & 1 & 0 \\
-1-\sqrt{2} & 2+\sqrt{2} & -\sqrt{17+13\sqrt{2}} & 1 & 0 & 0 \\
-2\sqrt{103+73\sqrt{2}} & 2\sqrt{41+29\sqrt{2}} & 11+8\sqrt{2} & 0 & 0 & 0 \\
6+5\sqrt{2} & -5-3\sqrt{2} & -2\sqrt{2\left(7+5\sqrt{2}\right)} & 0 & 1 & 0 \\
-7-5\sqrt{2} & 4+3\sqrt{2} & 2\sqrt{2\left(7+5\sqrt{2}\right)} & 1 & 0 & 0 \\
-\sqrt{-1+\sqrt{2}} & 0 & -1 & \sqrt{1+\sqrt{2}} & -\sqrt{2\left(1+\sqrt{2}\right)} & 0 \\
1-\sqrt{2} & -1 & 0 & 1+\sqrt{2} & -1-2\sqrt{2} & 0 \\
0 & 1 & 0 & 6+5\sqrt{2} & -5-3\sqrt{2} & 0 \\
1 & 0 & 0 & -7-5\sqrt{2} & 4+3\sqrt{2} & 0 \\
0 & 0 & -1 & 2\sqrt{103+73\sqrt{2}} & -2\sqrt{41+29\sqrt{2}} & 0 \\
0 & 0 & 0 & 0 & 0 & -11-8\cdots \\
0 & 0 & 0 & 0 & 0 & 2\sqrt{2\,(7+5\cdots} \\
0 & 0 & 0 & 0 & 0 & -2\sqrt{2\,(7+\cdots} \\
0 & 0 & 0 & 0 & 0 & -\sqrt{7+5\cdots} \\
0 & 0 & 0 & 0 & 0 & -\sqrt{17+13\cdots} \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & -1 \\
0 & 0 & 0 & 0 & 0 & 1 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0
\end{array}\right.
$$

$$
\begin{matrix}
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0
\end{matrix}
$$

In[141]:=
```
ArvMatPermB1 = ArvMatPerm[[1 ;; 10, 1 ;; 5]];
ArvMatPermB2 = ArvMatPerm[[11 ;; 20, 6 ;; 9]];
ArvMatPermB3 = ArvMatPerm[[21 ;; 30, 10 ;; 14]];
ArvMatPermB4 = ArvMatPerm[[31 ;; 40, 15 ;; 18]];
```

In[145]:=
```
ArvMatPermB1 // MatrixForm
ArvMatPermB2 // MatrixForm
ArvMatPermB3 // MatrixForm
ArvMatPermB4 // MatrixForm
```

Out[145]//MatrixForm=

$$
\begin{pmatrix}
-\sqrt{2} & 1+\sqrt{2} & -\sqrt{7+5\sqrt{2}} & 0 & 1 \\
-1-\sqrt{2} & 2+\sqrt{2} & -\sqrt{17+13\sqrt{2}} & 1 & 0 \\
-2\sqrt{103+73\sqrt{2}} & 2\sqrt{41+29\sqrt{2}} & 11+8\sqrt{2} & 0 & 0 \\
6+5\sqrt{2} & -5-3\sqrt{2} & -2\sqrt{2\left(7+5\sqrt{2}\right)} & 0 & 1 \\
-7-5\sqrt{2} & 4+3\sqrt{2} & 2\sqrt{2\left(7+5\sqrt{2}\right)} & 1 & 0 \\
-\sqrt{-1+\sqrt{2}} & 0 & -1 & \sqrt{1+\sqrt{2}} & -\sqrt{2\left(1+\sqrt{2}\right)} \\
1-\sqrt{2} & -1 & 0 & 1+\sqrt{2} & -1-2\sqrt{2} \\
0 & 1 & 0 & 6+5\sqrt{2} & -5-3\sqrt{2} \\
1 & 0 & 0 & -7-5\sqrt{2} & 4+3\sqrt{2} \\
0 & 0 & -1 & 2\sqrt{103+73\sqrt{2}} & -2\sqrt{41+29\sqrt{2}}
\end{pmatrix}
$$

Out[146]//MatrixForm=

$$\begin{pmatrix} -11-8\sqrt{2} & -2\sqrt{103+73\sqrt{2}} & 2\sqrt{41+29\sqrt{2}} & 1 \\ 2\sqrt{2\left(7+5\sqrt{2}\right)} & 6+5\sqrt{2} & -5-3\sqrt{2} & 0 \\ -2\sqrt{2\left(7+5\sqrt{2}\right)} & -7-5\sqrt{2} & 4+3\sqrt{2} & 0 \\ -\sqrt{7+5\sqrt{2}} & \sqrt{2} & -1-\sqrt{2} & \sqrt{-1+\sqrt{2}} \\ -\sqrt{17+13\sqrt{2}} & 1+\sqrt{2} & -2-\sqrt{2} & -\sqrt{1+\sqrt{2}} \\ 0 & 0 & -1 & 2\sqrt{2\left(7+5\sqrt{2}\right)} \\ 0 & -1 & 0 & -2\sqrt{2\left(7+5\sqrt{2}\right)} \\ -1 & 0 & 0 & 11+8\sqrt{2} \\ 1 & -\sqrt{-1+\sqrt{2}} & 0 & 1+2\sqrt{2} \\ 0 & 1-\sqrt{2} & -1 & 4\sqrt{1+\sqrt{2}} \end{pmatrix}$$

Out[147]//MatrixForm=

$$\begin{pmatrix} -6-5\sqrt{2} & 5+3\sqrt{2} & 2\sqrt{2\left(7+5\sqrt{2}\right)} & 0 & 1 \\ 7+5\sqrt{2} & -4-3\sqrt{2} & -2\sqrt{2\left(7+5\sqrt{2}\right)} & 1 & 0 \\ -2\sqrt{103+73\sqrt{2}} & 2\sqrt{41+29\sqrt{2}} & 11+8\sqrt{2} & 0 & 0 \\ \sqrt{1+\sqrt{2}} & -\sqrt{2\left(1+\sqrt{2}\right)} & 1+2\sqrt{2} & \sqrt{-1+\sqrt{2}} & 0 \\ 1+\sqrt{2} & -1-2\sqrt{2} & 4\sqrt{1+\sqrt{2}} & -1+\sqrt{2} & 1 \\ 0 & 0 & -1 & -2\sqrt{103+73\sqrt{2}} & 2\sqrt{41+29\sqrt{2}} \\ 0 & -1 & 0 & 6+5\sqrt{2} & -5-3\sqrt{2} \\ -1 & 0 & 0 & -7-5\sqrt{2} & 4+3\sqrt{2} \\ 0 & 1 & -\sqrt{-1+\sqrt{2}} & \sqrt{2} & -1-\sqrt{2} \\ 1 & 0 & \sqrt{1+\sqrt{2}} & 1+\sqrt{2} & -2-\sqrt{2} \end{pmatrix}$$

Out[148]//MatrixForm=

$$\begin{pmatrix} -1-2\sqrt{2} & \sqrt{1+\sqrt{2}} & -\sqrt{2\left(1+\sqrt{2}\right)} & 1 \\ -4\sqrt{1+\sqrt{2}} & 1+\sqrt{2} & -1-2\sqrt{2} & 0 \\ 2\sqrt{2\left(7+5\sqrt{2}\right)} & 6+5\sqrt{2} & -5-3\sqrt{2} & 0 \\ -2\sqrt{2\left(7+5\sqrt{2}\right)} & -7-5\sqrt{2} & 4+3\sqrt{2} & 0 \\ 11+8\sqrt{2} & 2\sqrt{103+73\sqrt{2}} & -2\sqrt{41+29\sqrt{2}} & 1 \\ -\sqrt{-1+\sqrt{2}} & 0 & -1 & -\sqrt{7+5\sqrt{2}} \\ \sqrt{1+\sqrt{2}} & -1 & 0 & -\sqrt{17+13\sqrt{2}} \\ 1 & 0 & 0 & 11+8\sqrt{2} \\ 0 & 0 & -1 & -2\sqrt{2\left(7+5\sqrt{2}\right)} \\ 0 & -1 & 0 & 2\sqrt{2\left(7+5\sqrt{2}\right)} \end{pmatrix}$$

We verify that we have obtained the correct blocks by checking that ArvMatPerm is equal to the direct sum of the blocks.

In[149]:=
```
ArvMatPerm == DirectSum[{ArvMatPermB1, ArvMatPermB2, ArvMatPermB3, ArvMatPermB4}]
```

Out[149]= True

Now for each block we compute the determinant of the submatrix formed from the first three rows. In all cases, the determinant is nonzero, hence each block has full rank, so ArvMat indeed has full rank, which completes the proof. Note that the blocks have different dimensions now, so the submatrix size that we need to take is different for different blocks.

In[150]:=
```
Map[Dimensions, {ArvMatPermB1, ArvMatPermB2, ArvMatPermB3, ArvMatPermB4}]
```

Out[150]= {{10, 5}, {10, 4}, {10, 5}, {10, 4}}

In[151]:=
```
Det[ArvMatPermB1[[1 ;; 5]]] // FullSimplify
Det[ArvMatPermB2[[1 ;; 4]]] // FullSimplify
Det[ArvMatPermB3[[1 ;; 5]]] // FullSimplify
Det[ArvMatPermB4[[1 ;; 4]]] // FullSimplify
```

Out[151]= $-16\left(7+5\sqrt{2}\right)$

Out[152]= $-4\sqrt{478+338\sqrt{2}}$

Out[153]= $16\sqrt{41+29\sqrt{2}}$

Out[154]= $-16\sqrt{82+58\sqrt{2}}$